Example nº 9
Design of the layout of a prestressing tendon

CivilFEM Manual of Advanced Examples Example nº 9 – Table of Contents

9 EXAMPLE Nº 9: DESIGN OF THE LAYOUT OF A PRESTRESSING TENDON

9.1 AIM

The aim of this example, that can be solved analytically, is to design the layout of a prestressed tendon of and to study the cracking of the structure.

9.2 DESCRIPTION OF THE EXAMPLE

For the continuous beam shown in the figure



a prestressing system has been designed formed by a polygonal line of straight segments, defined by the parameters e_1 and e_2, that fixes the position of the vertex, with the following values of the parameters that define the problem:

Parameter	Value	Units
L	10	m
h	100	cm
b	40	cm
f_{ck}	35	MPa
E_c	35000	MPa

9.3 RESULTS TO BE OBTAINED

Disregarding short term and long term losses, it is requested:

1. Study the secondary bending moment at the middle support for e_1 = 0.40 h and e_1/e_2 between 0.70 and 1.00.
2. Verify that for the ratio e_1/e_2 = 5/6 the previous secondary bending moment is null.
3. For e_1 = 0.40 h and e_1/e_2 = 1, a prestressing load of P = 1666.67 kN and considering the self weight of the structure, study the structure for values of Q between 400 kN and 600 kN and verify that the first crack appears at the upper fiber of the support section, for approximately Q = 455 kN (consider that a cracking takes place when the tension stress in the concrete is greater than $f_{ct,k}$ = 3 MPa).
4. Compare the analysis made in the previous section, with a decompression checking of the concrete (it is considered that cracking takes place when tension stresses appear in the concrete).

9.4 ANALYTIC RESOLUTION

9.4.1 Section 1

Taking e_1 and e_2 measured from the center of gravity of the section and positives upwards, the functions that describes the geometry of the prestressing tendon is:

$$\varphi_1(x) = \begin{cases} 2 \cdot \dfrac{e_1}{L} \cdot x & 0 \le x \le \dfrac{L}{2} \\ 2 \cdot \dfrac{e_2 - e_1}{L} \cdot x + 2 \cdot e_1 - e_2 & \dfrac{L}{2} \le x \le L \end{cases} \quad \text{for the left span}$$

(x is measured from the left support towards the right)

$$\varphi_2(x)=\begin{cases}\dfrac{e_1-e_2}{L}\cdot x+e_2 & 0\le x\le L\\ -\dfrac{e_1}{L}\cdot x+2\cdot e_1 & L\le x\le 2\cdot L\end{cases}\qquad \text{for the right span}$$

(x is measured from the central support towards the right)

The rotation caused by prestressing on the left span, at the section of the intermediate support is

$$\int_0^{L/2}\frac{P\cdot\varphi_1(x)}{E\cdot I}\cdot\frac{x}{L}\cdot dx+\int_{L/2}^{L}\frac{P\cdot\varphi_1(x)}{E\cdot I}\cdot\frac{x}{L}\cdot dx=$$

$$=\int_0^{L/2}\frac{P}{E\cdot I}\cdot\left(2\cdot\frac{e_1\cdot x^2}{L^2}\right)\cdot dx+\int_{L/2}^{L}\frac{P}{E\cdot I}\cdot\frac{x}{L}\cdot\left(2\cdot\frac{e_2-e_1}{L}\cdot x+2\cdot e_1-e_2\right)\cdot dx=$$

$$=\frac{P\cdot L}{E\cdot I}\left(\frac{e_1}{4}+\frac{5\cdot e_2}{24}\right)$$

The rotation at the same section, but on the right span is

$$\int_0^{L}\frac{P\cdot\varphi_2(x)}{E\cdot I}\cdot\left(L-\frac{x}{2\cdot L}\right)\cdot dx+\int_{L}^{2L}\frac{P\cdot\varphi_2(x)}{E\cdot I}\cdot\left(L-\frac{x}{2\cdot L}\right)\cdot dx=$$

$$=\int_0^{L}\frac{P}{E\cdot I}\cdot\left(\frac{e_1-e_2}{L}\cdot x+e_2\right)\cdot\left(L-\frac{x}{2\cdot L}\right)\cdot dx+\int_{L}^{2L}\frac{P}{E\cdot I}\cdot\frac{x}{L}\cdot\left(-\frac{e_1}{L}\cdot x+2\cdot e_1\right)\cdot\left(L-\frac{x}{2\cdot L}\right)\cdot dx=$$

$$=\frac{P\cdot L}{E\cdot I}\left(\frac{e_1}{2}+\frac{5\cdot e_2}{12}\right)$$

Forcing for the rotations to be equal, a secondary bending moment appears:

$$M_H=-P\cdot\left(\frac{3}{4}e_1+\frac{5}{8}e_2\right)$$

9.4.2 Section 2

Taking e_1/e_2 = 5/6, with the additional precaution of changing the sign of e_1, the previously calculated secondary moment is null.

(the sign change for e_1 is done because for the calculation of M_H the positive displacement of the tendon is assumed upwards, whereas the example description considers e_1 positive downwards).

9.4.3 Section 3

The stress at the upper fibers of the section at the intermediate support is given by the expression:

$$\sigma = \frac{P}{A} + \frac{M_P + M_W}{I} \cdot \frac{h}{2}$$

where M_P is the bending moment produced by prestressing and punctual loads and M_W the bending moment produced by self weight.

In this case, using Q = 455 kN:

$$M_P = \frac{3}{16} \cdot (Q - 400) \cdot L = 103.125 \text{ kN}$$
$$M_W = -0.375 \cdot \gamma_c \cdot A \cdot L^2 = 368.565 \text{ kN}$$

And the stress is therefore

$$\sigma = -\frac{1666.67}{0.4} + \frac{103.125 + 368.565}{0.0333} \cdot \frac{1.0}{2} = 2915.76 \text{ kN} \approx 3000 \text{ kN}$$

which will produce the first crack, as it was indicated in the statement of the exercise.

9.5 EXECUTION LOG

9.5.1 Introduction

The units used in the model are meter, second and kiloNewton, and the derived ones.

The analysis uses a series of definition parameters for the model. Two of these parameters are e_1 and e_2. For the first section, the example is executed several times, changing the values e_2, thus to construct the graph shown in the results. One of the chosen values, which is the one used in the log file of this example, is e_2/e_1 = 5/6, that answers section 2 of the statement.

The prestressing force P is defined in sections 3 and 4. As the secondary bending moments law which must be obtained in sections 1 and 2 is proportional to P, and the desired result is to achieve a null secondary bending moment, the results will be the same for any value of P. For the resolution of sections 1 and 2, a value of P = 4000 kN has been used.

Load Q used in sections 3 and 4 is null in both first sections, but the *log* file includes it in the parameters list in order to be able to use the same file in all the cases, only changing the values of the initial parameters.

The materials are taken from CivilFEM library, choosing a HA-35 of the Spanish concrete code EHE. This concrete has a strength of 35 MPa when it is 28 days old as it is specified in the statement. For the elasticity modulus to also correspond with the one in the example data, the initial modulus type is selected.

The election of one or other prestressing steel material does not affect the results, because all the parameters of short term losses are set to zero, and long term losses will not be considered in the analysis.

After choosing the materials *a macro* that contains the generation of the model, based on all the defined parameters, is executed. The model will have 60 elements and 61 nodes.

The complete model is then captured for the creation of the support beam, on which the tendon is defined, from the given geometric description in the statement.

The slopes of the tendon must be continuous. This is why, to define the layout of the example, it is necessary to smooth the angles of the tendon.

For it, the cuts and the points of the vertices must be duplicated. This way it is possible to introduce a small curve in where a jump in the slope would have to take place.



Before solving, acceleration should be entered, so that the analysis considers the structure self weight (only for sections 3 and 4).

The obtained results are:

- Bending moment’s law (used in sections 1 and 2).
- Minimum stress tension at each section (used in section 3).
- Decompression check and plot of tensioned elements (used in section 4).

9.5.2 Model generation

```
FINISH
~CFCLEAR,,1

~CODESEL,,EHE,EHE
~UNITS,,LENG,M
~UNITS,,TIME,S
~UNITS,,FORC,KN

/PREP7
! Parameters
L  = 10  ! Beam length (m)
b  = 0.4 ! Section width (m)
h  = 1.0 ! Section height (m)
e1 = 0.4*h    ! Tendon position (m)
e2 = e1*5/6   ! Tendon position (m)
Q  = 0     ! External load (kN)
P  = 2000 ! Prestressing load (kN)

! Material definition
~CFMP,1,LIB,CONCRETE,EHE,HA-35
~CFMP,1,Concr,TPEX,,2
~CFMP,2,LIB,PREST,EHE,Y1860S7
~CFMP,2,Prest,A     ,,0.0
~CFMP,2,Prest,MU    ,,0.0
~CFMP,2,Prest,K     ,,0.0

! Beam properties definition
~CSECDMS,1,REC,1,h,b
~BMSHPRO,1,BEAM,1,1,,,44,1,1,Beam 1

! Nodes
Epsilon = 0.05
N,1                  ! Initial node
N,11,L/2             ! Node at L/2
N,12,L/2+Epsilon     ! Node at the left of L/2
N,21,L               ! Node at L
N,22,L+Epsilon       ! Node at the left of L
N,41,2*L             ! Node at 2*L
N,42,2*L+Epsilon     ! Node at the left of 2*L
N,61,3*L             ! Node at 3*L
FILL,1,11 $ FILL,12,21 $ FILL,22,41 $ FILL,42,61

! Elements
MAT,1 $ ET,1,BEAM44 $ TYPE,1 $ REAL,1
E,1,2
EGEN,60,1,1,2,1

! Boundary conditions
D,1,UY  $ D,1,UZ
```

```
D,20,UX $ D,20,UY $ D,20,UZ $ D,20,ROTX
D,61,UY $ D,61,UZ

! Support Beam
~SBBMDEF

! Tendon Geometry
~PCEPDEF,1,1,0
~PCEPDEF,2,11,-e1
~PCEPDEF,3,12,-e1
~PCEPDEF,4,21,e2
~PCEPDEF,5,22,e2
~PCEPDEF,6,41,-e1
~PCEPDEF,7,42,-e1
~PCEPDEF,8,61,0
~PCPPDEF,1,1,0,0
~PCPPDEF,2,61,0,0
~PCTNDEF,1,2,10e-4,0.001,P,P
~PCTNMDF,1,EADD,1
~PCTNMDF,1,EADD,2
~PCTNMDF,1,EADD,3
~PCTNMDF,1,EADD,4
~PCTNMDF,1,EADD,5
~PCTNMDF,1,EADD,6
~PCTNMDF,1,EADD,7
~PCTNMDF,1,EADD,8
~PCTNMDF,1,PADD,1
~PCTNMDF,1,PADD,2
~PCTNMDF,1,ESTRLN,1,0
~PCTNMDF,1,ESTRLN,3,0
~PCTNMDF,1,ESTRLN,5,0
~PCTNMDF,1,ESTRLN,7,0

! Losses Calculation
~PCLOSS,0,0,,0

! Apply prestressing forces
~PCPL

! Apply loads
FCUM,ADD
F,11,FY,-Q
F,41,FY,-Q/2

! Active time = 28 days
~ACTTIME,28
```



9.5.3 Analysis

```
/SOLU

ACEL,,9.81 ! Used in questions 3 and 4 only.
           ! For questions 1 and 2 it must be removed.

SOLVE
```

9.5.4 Postprocess

```
/POST1
! Point to result load state
~CFSET,,1

! Questions 1 and 2. Bending moment distribution.
! Plot results
~PLLSFOR,HM,Z,-1

! Question 3. Maximum stress at each section.
~PLLSSTR,SX,MAX

! Question 4. Decompression check.
~CHKPRS,CRACK
~PLLSPRS,ELM_OK
~PLLSPRS,CRT_TOT
```

9.6 RESULTS

9.6.1 Section 1

Using the resolution explained in the previous section, and giving the parameter e_2 different values to vary the ratio e_1/e_2 the following values of the secondary bending moment at the intermediate support have been obtained (for a prestressing force of 2000 kN):

e_1/e_2	$M_{Hyp,Z}$ (kN·m)
0.70	-109.148
0.75	-62.257
0.80	-21.165
0.85	15.138
0.90	47.442
0.95	76.372
1.00	102.429



9.6.2 Section 2

For the value $\frac{e_1}{e_2} = \frac{5}{6} = 0{,}833$ a secondary bending moment is obtained at the support of 3.516 kN·m. It is not strictly null because of the differences between the model and the analytical solution: In the model the angular points of the tendon are smoothed by means of a small curve and in the theoretical model simplifications are made to apply the prestressing loads considering small angles in the tendon's geometry definition.

9.6.3 Section 3

For the different values of Q the following maximum stresses in the model have been obtained. It is important to consider that the tensions are represented by positive stresses.



The maximum stress is not always obtained at the same point of the section.

For the value of Q = 455 kN a maximum stress of 3.076 MPa is obtained (2.5% of difference with respect to the $f_{ct,k}$ = 3 MPa strength). The location of this tension is indeed the upper fiber of the section, as it can be seen in the graphical representation of stresses:



9.6.4 Section 4

Concrete decompression has been studied. The results are shown in the following figure:



As it is normal, the decompression limit of concrete (null stress) is reached before (lower values of Q) than the tension characteristic of traction.

As it can be seen in the following figure (elements that fulfill or not the checking criteria), there is an element in which decompression takes place for Q = 350 kN and is located on the central support.



Decompression check for Q = 350 kN

9.7 SUMMARY

In this exercise, the reader has been able to analyze a real case in which he is to obtain, by means of trial and check, the geometry the tendon must have.

Another additional difficulty has been introduced in this example, which is the existence of angular points in the layout of the tendon. As it has been seen, this must be dealt in a particular way.